Expansions based on Johns, Ellis and Lattimer

```
Off[General::"spell"]; Off[General::"spell1"];
```

- **Non-degenerate and non-relativstic:**

```
pcoeff = t^2 (-1)^(n+1) Exp[n (ψ + 1/t)] / n^2
```

$$\frac{(-1)^{1+n}\, e^{n\left(\frac{1}{t}+\psi\right)}\, t^2}{n^2}$$

```
besselseries = Simplify[Normal[Series[BesselK[2, x], {x, ∞, 6}]]]
```

$$\frac{1}{4\,194\,304}\, e^{-x} \sqrt{\frac{\pi}{2}} \left(\frac{1}{x}\right)^{13/2} \left(4\,729\,725 - 2\,162\,160\, x + 1\,330\,560\, x^2 - 1\,290\,240\, x^3 + 3\,440\,640\, x^4 + 7\,864\,320\, x^5 + 4\,194\,304\, x^6\right)$$

```
pNDNR = Simplify[pcoeff (besselseries /. x → n / t)]
```

$$-\frac{1}{4\,194\,304\, n^6} (-1)^n e^{n\psi} \sqrt{\frac{\pi}{2}} \left(\frac{t}{n}\right)^{5/2} \left(4\,194\,304\, n^6 + 7\,864\,320\, n^5\, t + 3\,440\,640\, n^4\, t^2 - 1\,290\,240\, n^3\, t^3 + 1\,330\,560\, n^2\, t^4 - 2\,162\,160\, n\, t^5 + 4\,729\,725\, t^6\right)$$

Separate out the first term to make the expansion more clear:

```
firstterm = (Simplify[pNDNR / t^(5/2), t > 0] /. t → 0) t^(5/2)
```

$$-(-1)^n e^{n\psi} \left(\frac{1}{n}\right)^{5/2} \sqrt{\frac{\pi}{2}}\, t^{5/2}$$

```
pNDNRsimp = Expand[Simplify[{firstterm, pNDNR / firstterm}, n > 0]]
```

$$\left\{-\frac{(-1)^n e^{n\psi} \sqrt{\frac{\pi}{2}}\, t^2 \sqrt{\frac{t}{n}}}{n^2},\ 1 + \frac{15\, t}{8\, n} + \frac{105\, t^2}{128\, n^2} - \frac{315\, t^3}{1024\, n^3} + \frac{10\,395\, t^4}{32\,768\, n^4} - \frac{135\,135\, t^5}{262\,144\, n^5} + \frac{4\,729\,725\, t^6}{4\,194\,304\, n^6}\right\}$$

Calculate the number and energy densities:

```
ρNDNR = D[pNDNR, ψ] / t
```

$$-\frac{1}{4\,194\,304\, n^5\, t} (-1)^n e^{n\psi} \sqrt{\frac{\pi}{2}} \left(\frac{t}{n}\right)^{5/2} \left(4\,194\,304\, n^6 + 7\,864\,320\, n^5\, t + 3\,440\,640\, n^4\, t^2 - 1\,290\,240\, n^3\, t^3 + 1\,330\,560\, n^2\, t^4 - 2\,162\,160\, n\, t^5 + 4\,729\,725\, t^6\right)$$

```
ft2 = (Simplify[ρNDNR / t^(3/2), t > 0] /. t → 0) t^(3/2);
ρNDNRsimp = Expand[Simplify[{ft2, ρNDNR / ft2}, t > 0]]
```

$$\left\{-\frac{(-1)^n e^{n\psi} \sqrt{\frac{\pi}{2}}\, t \sqrt{\frac{t}{n}}}{n},\ 1 + \frac{15\, t}{8\, n} + \frac{105\, t^2}{128\, n^2} - \frac{315\, t^3}{1024\, n^3} + \frac{10\,395\, t^4}{32\,768\, n^4} - \frac{135\,135\, t^5}{262\,144\, n^5} + \frac{4\,729\,725\, t^6}{4\,194\,304\, n^6}\right\}$$

```
εNDNR = Simplify[t D[pNDNR, t] - pNDNR + ρNDNR]
```

$$-\frac{1}{8\,388\,608\,n^7}\left((-1)^n e^{n\psi}\sqrt{\frac{\pi}{2}}\left(\frac{t}{n}\right)^{3/2}\left(8\,388\,608\,n^7+28\,311\,552\,n^6\,t+46\,202\,880\,n^5\,t^2+21\,504\,000\,n^4\,t^3-8\,951\,040\,n^3\,t^4+10\,311\,840\,n^2\,t^5-18\,648\,630\,n\,t^6+70\,945\,875\,t^7\right)\right)$$

```
ft3 = (Simplify[εNDNR/t^(3/2), t > 0] /. t → 0) t^(3/2);
εNDNRsimp = Expand[Simplify[{ft3, εNDNR/ft3}, t > 0]]
```

$$\left\{-\frac{(-1)^n e^{n\psi}\sqrt{\frac{\pi}{2}}\,t\,\sqrt{\frac{t}{n}}}{n},\ 1+\frac{27\,t}{8\,n}+\frac{705\,t^2}{128\,n^2}+\frac{2625\,t^3}{1024\,n^3}-\frac{34\,965\,t^4}{32\,768\,n^4}+\frac{322\,245\,t^5}{262\,144\,n^5}-\frac{9\,324\,315\,t^6}{4\,194\,304\,n^6}+\frac{70\,945\,875\,t^7}{8\,388\,608\,n^7}\right\}$$

■ Non-degenerate and extremely relativistic:

```
pcoeff = t^2 (-1)^(n+1) Exp[n ψ] / n^2
```

$$\frac{(-1)^{1+n} e^{n\psi}\,t^2}{n^2}$$

```
bs = Series[BesselK[2, x] Exp[x], {x, 0, 3}]
```

$$\frac{2}{x^2}+\frac{2}{x}+\frac{1}{2}-\frac{x}{6}+\left(-\frac{7}{96}-\frac{\text{EulerGamma}}{8}+\frac{\text{Log}[2]}{8}-\frac{\text{Log}[x]}{8}\right)x^2+\left(\frac{13}{480}-\frac{\text{EulerGamma}}{8}+\frac{\text{Log}[2]}{8}-\frac{\text{Log}[x]}{8}\right)x^3+O[x]^4$$

Separate the individual terms:

```
bs2 = {Normal[Series[BesselK[2, x] Exp[x], {x, 0, 1}]],
  SeriesCoefficient[bs, 2], SeriesCoefficient[bs, 3]}
```

$$\left\{\frac{1}{2}+\frac{2}{x^2}+\frac{2}{x}-\frac{x}{6},\ -\frac{7}{96}-\frac{\text{EulerGamma}}{8}+\frac{\text{Log}[2]}{8}-\frac{\text{Log}[x]}{8},\ \frac{13}{480}-\frac{\text{EulerGamma}}{8}+\frac{\text{Log}[2]}{8}-\frac{\text{Log}[x]}{8}\right\}$$

```
pNDER = Simplify[pcoeff (bs2 /. x → n / t)]
```

$$\left\{\frac{(-1)^n e^{n\psi}\,t\left(n^3-3\,n^2\,t-12\,n\,t^2-12\,t^3\right)}{6\,n^4},\ \frac{(-1)^n e^{n\psi}\,t^2\left(7+12\,\text{EulerGamma}-12\,\text{Log}[2]+12\,\text{Log}\left[\frac{n}{t}\right]\right)}{96\,n^2},\ \frac{(-1)^n e^{n\psi}\,t^2\left(-13+60\,\text{EulerGamma}-60\,\text{Log}[2]+60\,\text{Log}\left[\frac{n}{t}\right]\right)}{480\,n^2}\right\}$$

```
coefx = -2 t^4 Exp[n ψ] (-1)^n / n^4
```

$$-\frac{2\,(-1)^n e^{n\psi}\,t^4}{n^4}$$

```
pNDERsimp = Expand[{coefx, pNDER / coefx}]
```

$$\left\{-\frac{2\,(-1)^n\,e^{n\psi}\,t^4}{n^4},\ \left\{1-\frac{n^3}{12\,t^3}+\frac{n^2}{4\,t^2}+\frac{n}{t},\ -\frac{7\,n^2}{192\,t^2}-\frac{\text{EulerGamma}\,n^2}{16\,t^2}+\frac{n^2\,\text{Log}[2]}{16\,t^2}-\frac{n^2\,\text{Log}\left[\frac{n}{t}\right]}{16\,t^2},\ \frac{13\,n^2}{960\,t^2}-\frac{\text{EulerGamma}\,n^2}{16\,t^2}+\frac{n^2\,\text{Log}[2]}{16\,t^2}-\frac{n^2\,\text{Log}\left[\frac{n}{t}\right]}{16\,t^2}\right\}\right\}$$

Compute the number and energy densities:

```
ρNDER = D[pNDER, ψ] / t
```

$$\left\{\frac{(-1)^n\,e^{n\psi}\left(n^3-3\,n^2\,t-12\,n\,t^2-12\,t^3\right)}{6\,n^3},\ \frac{(-1)^n\,e^{n\psi}\,t\left(7+12\,\text{EulerGamma}-12\,\text{Log}[2]+12\,\text{Log}\left[\frac{n}{t}\right]\right)}{96\,n},\ \frac{(-1)^n\,e^{n\psi}\,t\left(-13+60\,\text{EulerGamma}-60\,\text{Log}[2]+60\,\text{Log}\left[\frac{n}{t}\right]\right)}{480\,n}\right\}$$

```
coefx2 = -2 t^3 Exp[n ψ] (-1)^n / n^3
```

$$-\frac{2\,(-1)^n\,e^{n\psi}\,t^3}{n^3}$$

```
ρNDERsimp = {coefx2, Expand[ρNDER / coefx2]}
```

$$\left\{-\frac{2\,(-1)^n\,e^{n\psi}\,t^3}{n^3},\ \left\{1-\frac{n^3}{12\,t^3}+\frac{n^2}{4\,t^2}+\frac{n}{t},\ -\frac{7\,n^2}{192\,t^2}-\frac{\text{EulerGamma}\,n^2}{16\,t^2}+\frac{n^2\,\text{Log}[2]}{16\,t^2}-\frac{n^2\,\text{Log}\left[\frac{n}{t}\right]}{16\,t^2},\ \frac{13\,n^2}{960\,t^2}-\frac{\text{EulerGamma}\,n^2}{16\,t^2}+\frac{n^2\,\text{Log}[2]}{16\,t^2}-\frac{n^2\,\text{Log}\left[\frac{n}{t}\right]}{16\,t^2}\right\}\right\}$$

```
εNDER = Simplify[t D[pNDER, t] - pNDER + ρNDER]
```

$$\left\{\frac{(-1)^n\,e^{n\psi}\left(n^4-3\,n^3\,t-15\,n^2\,t^2-36\,n\,t^3-36\,t^4\right)}{6\,n^4},\ \frac{1}{96\,n^2}\,(-1)^n\,e^{n\psi}\,t\left(t\,(-5+12\,\text{EulerGamma}-12\,\text{Log}[2])+n\,(7+12\,\text{EulerGamma}-12\,\text{Log}[2])+12\,(n+t)\,\text{Log}\left[\frac{n}{t}\right]\right),\ \frac{1}{480\,n^2}\,(-1)^n\,e^{n\psi}\,t\left(t\,(-73+60\,\text{EulerGamma}-60\,\text{Log}[2])+n\,(-13+60\,\text{EulerGamma}-60\,\text{Log}[2])+60\,(n+t)\,\text{Log}\left[\frac{n}{t}\right]\right)\right\}$$

```
coefx3 = -6 t^3 Exp[n ψ] (-1)^n / n^3
```

$$-\frac{6\,(-1)^n\,e^{n\psi}\,t^3}{n^3}$$

```
εNDERsimp = {coefx3, Expand[εNDER / coefx3]}
```

$$\left\{-\frac{6\,(-1)^n\,e^{n\psi}\,t^3}{n^3},\ \left\{1-\frac{n^3}{36\,t^3}+\frac{n^2}{12\,t^2}+\frac{5\,n}{12\,t}+\frac{t}{n},\right.\right.$$
$$-\frac{7\,n^2}{576\,t^2}-\frac{\text{EulerGamma}\,n^2}{48\,t^2}+\frac{5\,n}{576\,t}-\frac{\text{EulerGamma}\,n}{48\,t}+\frac{n^2\,\text{Log}[2]}{48\,t^2}+\frac{n\,\text{Log}[2]}{48\,t}-\frac{n^2\,\text{Log}\left[\frac{n}{t}\right]}{48\,t^2}-\frac{n\,\text{Log}\left[\frac{n}{t}\right]}{48\,t},$$
$$\left.\left.\frac{13\,n^2}{2880\,t^2}-\frac{\text{EulerGamma}\,n^2}{48\,t^2}+\frac{73\,n}{2880\,t}-\frac{\text{EulerGamma}\,n}{48\,t}+\frac{n^2\,\text{Log}[2]}{48\,t^2}+\frac{n\,\text{Log}[2]}{48\,t}-\frac{n^2\,\text{Log}\left[\frac{n}{t}\right]}{48\,t^2}-\frac{n\,\text{Log}\left[\frac{n}{t}\right]}{48\,t}\right\}\right\}$$

▪ The extremely degenerate case:

The pressure from JEL:

```
p = 1 / 3 Integrate[f[l] l^4 / Sqrt[l^2 + 1] / (1 + Exp[(Sqrt[l^2 + 1] - 1) / t - ψ]), {l, 0, ∞}]; p /. f[l] → 1
```

$$\frac{1}{3}\int_0^{\infty}\frac{l^4}{\left(1+e^{\frac{-1+\sqrt{1+l^2}}{t}-\psi}\right)\sqrt{1+l^2}}\,dl$$

Change variables to $l=\sqrt{(z+1)^2-1}$ and ψ t=x, and re-examine the integrand:

```
dldz = D[Sqrt[(z + 1)^2 - 1], z]
```

$$\frac{1+z}{\sqrt{-1+(1+z)^2}}$$

```
f[l] dl == f[z] (dl / dz) dz
```

$$dl\,f[l]==dl\,f[z]$$

```
degiand = Simplify[
  l^4 dldz / 3 / Sqrt[l^2 + 1] / (1 + Exp[(Sqrt[l^2 + 1] - 1) / t - ψ]) /. l → Sqrt[(z + 1)^2 - 1] /. ψ → x / t,
  {z > 0}]
```

$$\frac{e^{\frac{x}{t}}\,(z\,(2+z))^{3/2}}{3\left(e^{\frac{x}{t}}+e^{\frac{z}{t}}\right)}$$

Note that the limits are the same.

Separate out the Fermi function:

```
newf = degiand / Exp[x / t] (e^(x/t) + e^(z/t))
```

$$\frac{1}{3}\,(z\,(2+z))^{3/2}$$

```
Simplify[degiand - newf / (1 + Exp[z / t - x / t])]
```

$$0$$

Now we can use the Sommerfeld expansion from Landau and Lifshitz, which is valid when μ /T(=x/t) is large.
Note that this is an asymptotic expansion, so we have to be careful about including too many terms.

```
eq = Integrate[f[ϵ] / (1 + Exp[(ϵ - μ) / T]), {ϵ, 0, ∞}] == Expand[Integrate[f[ϵ], {ϵ, 0, μ}] +
    Integrate[T Normal[Series[(f[μ + T z] - f[μ - T z]), {z, 0, 5}]] / (Exp[z] + 1), {z, 0, ∞}]]
```

$$\int_0^\infty \frac{f[\epsilon]}{1+e^{\frac{\epsilon-\mu}{T}}}\,d\epsilon == \int_0^\mu f[\epsilon]\,d\epsilon + \frac{1}{6}\pi^2 T^2 f'[\mu] + \frac{7}{360}\pi^4 T^4 f^{(3)}[\mu] + \frac{31\pi^6 T^6 f^{(5)}[\mu]}{15120}$$

We add coefficients $\alpha 1$ and $\alpha 2$ to see how the Sommerfeld expansion contributes
(we'll set them to one later)

```
ped = Simplify[Integrate[newf, {z, 0, x}] + α1 π^2 t^2 / 6 (D[newf, z] /. z → x) +
    7 α2 π^4 t^4 / 360 (D[D[D[newf, z], z], z] /. z → x), x > 0] /. x → ψ t
```

$$\frac{1}{360\,(t\psi(2+t\psi))^{3/2}}\Big((1+t\psi)\left(-7\pi^4 t^4 \alpha2 + 28\pi^4 t^5 \alpha2\,\psi + 2t^2\left(-90+120\pi^2 t^2\alpha1 + 7\pi^4 t^4\alpha2\right)\psi^2 + 60t^3\left(1+4\pi^2t^2\alpha1\right)\psi^3 + 15t^4\left(21+4\pi^2t^2\alpha1\right)\psi^4 + 180t^5\psi^5 + 30t^6\psi^6\right) + 90\left(2\sqrt{t^3\psi^3(2+t\psi)} + \sqrt{t^5\psi^5(2+t\psi)}\right)\text{ArcSinh}\left[\frac{\sqrt{t\psi}}{\sqrt{2}}\right]\Big)$$

- **The extremely degenerate and non-relativistic case:**

```
pEDNR = Expand[Simplify[Normal[Series[ped, {t, 0, 5}]], {ψ > 0}]]
```

$$-\frac{7\pi^4 t^{5/2}\alpha2}{720\sqrt{2}\,\psi^{3/2}} + \frac{7\pi^4 t^{7/2}\alpha2}{192\sqrt{2}\sqrt{\psi}} + \frac{\pi^2 t^{5/2}\alpha1\sqrt{\psi}}{3\sqrt{2}} + \frac{49\pi^4 t^{9/2}\alpha2\sqrt{\psi}}{1536\sqrt{2}} + \frac{5\pi^2 t^{7/2}\alpha1\,\psi^{3/2}}{12\sqrt{2}} + \frac{4}{15}\sqrt{2}\,t^{5/2}\psi^{5/2} + \frac{7\pi^2 t^{9/2}\alpha1\,\psi^{5/2}}{96\sqrt{2}} + \frac{1}{7}\sqrt{2}\,t^{7/2}\psi^{7/2} + \frac{t^{9/2}\psi^{9/2}}{36\sqrt{2}}$$

```
coefy = 4 Sqrt[2] ψ^5/2 t^5/2 / 15
```

$$\frac{4}{15}\sqrt{2}\,t^{5/2}\psi^{5/2}$$

```
pEDNRsimp = {coefy, Expand[pEDNR / coefy]}
```

$$\left\{\frac{4}{15}\sqrt{2}\,t^{5/2}\psi^{5/2},\ 1 + \frac{35}{256}\pi^2 t^2\alpha1 - \frac{7\pi^4\alpha2}{384\psi^4} + \frac{35\pi^4 t\,\alpha2}{512\psi^3} + \frac{5\pi^2\alpha1}{8\psi^2} + \frac{245\pi^4 t^2\alpha2}{4096\psi^2} + \frac{25\pi^2 t\,\alpha1}{32\psi} + \frac{15t\psi}{28} + \frac{5t^2\psi^2}{96}\right\}$$

There appears to be a typo in the reference?

Compute the number and energy densities:

ρEDNR = Expand[D[pEDNR, ψ] / t]

$$\frac{7\pi^4 t^{3/2}\alpha 2}{480\sqrt{2}\,\psi^{5/2}}-\frac{7\pi^4 t^{5/2}\alpha 2}{384\sqrt{2}\,\psi^{3/2}}+\frac{\pi^2 t^{3/2}\alpha 1}{6\sqrt{2}\sqrt{\psi}}+\frac{49\pi^4 t^{7/2}\alpha 2}{3072\sqrt{2}\sqrt{\psi}}+\frac{5\pi^2 t^{5/2}\alpha 1\sqrt{\psi}}{8\sqrt{2}}+\frac{2}{3}\sqrt{2}\,t^{3/2}\psi^{3/2}+\frac{35\pi^2 t^{7/2}\alpha 1\,\psi^{3/2}}{192\sqrt{2}}+\frac{t^{5/2}\psi^{5/2}}{\sqrt{2}}+\frac{t^{7/2}\psi^{7/2}}{8\sqrt{2}}$$

coefy2 = 2 Sqrt[2] / 3 t^{3/2} ψ^{3/2}

$$\frac{2}{3}\sqrt{2}\,t^{3/2}\psi^{3/2}$$

ρEDNRsimp = {coefy2, Expand[ρEDNR / coefy2]}

$$\left\{\frac{2}{3}\sqrt{2}\,t^{3/2}\psi^{3/2},\ 1+\frac{35}{256}\pi^2 t^2\alpha 1+\frac{7\pi^4\alpha 2}{640\psi^4}-\frac{7\pi^4 t\,\alpha 2}{512\psi^3}+\frac{\pi^2\alpha 1}{8\psi^2}+\frac{49\pi^4 t^2\alpha 2}{4096\psi^2}+\frac{15\pi^2 t\,\alpha 1}{32\psi}+\frac{3t\psi}{4}+\frac{3t^2\psi^2}{32}\right\}$$

εEDNR = Expand[Simplify[t D[pEDNR, t] - pEDNR + ρEDNR]]

$$\frac{7\pi^4 t^{3/2}\alpha 2}{480\sqrt{2}\,\psi^{5/2}}-\frac{21\pi^4 t^{5/2}\alpha 2}{640\sqrt{2}\,\psi^{3/2}}+\frac{\pi^2 t^{3/2}\alpha 1}{6\sqrt{2}\sqrt{\psi}}+\frac{329\pi^4 t^{7/2}\alpha 2}{3072\sqrt{2}\sqrt{\psi}}+\frac{9\pi^2 t^{5/2}\alpha 1\sqrt{\psi}}{8\sqrt{2}}+\frac{343\pi^4 t^{9/2}\alpha 2\sqrt{\psi}}{3072\sqrt{2}}+\frac{2}{3}\sqrt{2}\,t^{3/2}\psi^{3/2}+\frac{235\pi^2 t^{7/2}\alpha 1\,\psi^{3/2}}{192\sqrt{2}}+\frac{9t^{5/2}\psi^{5/2}}{5\sqrt{2}}+\frac{49\pi^2 t^{9/2}\alpha 1\,\psi^{5/2}}{192\sqrt{2}}+\frac{47t^{7/2}\psi^{7/2}}{56\sqrt{2}}+\frac{7t^{9/2}\psi^{9/2}}{72\sqrt{2}}$$

εEDNRsimp = {coefy2, Expand[εEDNR / coefy2]}

$$\left\{\frac{2}{3}\sqrt{2}\,t^{3/2}\psi^{3/2},\ 1+\frac{235}{256}\pi^2 t^2\alpha 1+\frac{7\pi^4\alpha 2}{640\psi^4}-\frac{63\pi^4 t\,\alpha 2}{2560\psi^3}+\frac{\pi^2\alpha 1}{8\psi^2}+\frac{329\pi^4 t^2\alpha 2}{4096\psi^2}+\frac{27\pi^2 t\,\alpha 1}{32\psi}+\frac{343\pi^4 t^3\alpha 2}{4096\psi}+\frac{27t\psi}{20}+\frac{49}{256}\pi^2 t^3\alpha 1\,\psi+\frac{141t^2\psi^2}{224}+\frac{7t^3\psi^3}{96}\right\}$$

▪ The extremely degenerate and extremely relativistic case:

p2 = Expand[Simplify[Normal[Series[ped, {t, ∞, -1}]], ψ > 0]]

$$\frac{1}{12}\pi^2 t^2\alpha 1+\frac{7}{180}\pi^4 t^4\alpha 2-\frac{t\psi}{6}+\frac{1}{3}\pi^2 t^3\alpha 1\,\psi+\frac{t^2\psi^2}{4}+\frac{1}{6}\pi^2 t^4\alpha 1\,\psi^2+\frac{t^3\psi^3}{3}+\frac{t^4\psi^4}{12}$$

p2big = Series[ped, {t, ∞, 2}];

**pEDER = Simplify[{Normal[Series[ped, {t, ∞, -1}]], SeriesCoefficient[p2big, 0],
SeriesCoefficient[p2big, 1], SeriesCoefficient[p2big, 2]}, ψ > 0]**

$$\left\{\frac{1}{180}t\left(7\pi^4 t^3\alpha 2+15\pi^2 t\,\alpha 1\left(1+4t\psi+2t^2\psi^2\right)+15\psi\left(-2+3t\psi+4t^2\psi^2+t^3\psi^3\right)\right),\ \frac{1}{480}\left(-35-\frac{7\pi^4\alpha 2}{\psi^4}-\frac{10\pi^2\alpha 1}{\psi^2}-60\,\mathrm{Log}\left[\frac{1}{t}\right]+60\,\mathrm{Log}[2\psi]\right),\ \frac{7\pi^4\alpha 2+5\pi^2\alpha 1\,\psi^2+15\psi^4}{120\psi^5},\ -\frac{7\left(7\pi^4\alpha 2+3\pi^2\alpha 1\,\psi^2+3\psi^4\right)}{288\psi^6}\right\}$$

```
coefz = ψ^4 t^4 / 12
```

$$\frac{t^4\,\psi^4}{12}$$

```
pEDERsimp = {coefz, Expand[pEDER / coefz]}
```

$$\left\{\frac{t^4\,\psi^4}{12},\ \left\{1+\frac{\pi^2\,\alpha1}{t^2\,\psi^4}+\frac{7\,\pi^4\,\alpha2}{15\,\psi^4}-\frac{2}{t^3\,\psi^3}+\frac{4\,\pi^2\,\alpha1}{t\,\psi^3}+\frac{3}{t^2\,\psi^2}+\frac{2\,\pi^2\,\alpha1}{\psi^2}+\frac{4}{t\,\psi},\right.\right.$$
$$-\frac{7\,\pi^4\,\alpha2}{40\,t^4\,\psi^8}-\frac{\pi^2\,\alpha1}{4\,t^4\,\psi^6}-\frac{7}{8\,t^4\,\psi^4}-\frac{3\,\mathrm{Log}\left[\frac{1}{t}\right]}{2\,t^4\,\psi^4}+\frac{3\,\mathrm{Log}[2\,\psi]}{2\,t^4\,\psi^4},$$
$$\left.\left.\frac{7\,\pi^4\,\alpha2}{10\,t^4\,\psi^9}+\frac{\pi^2\,\alpha1}{2\,t^4\,\psi^7}+\frac{3}{2\,t^4\,\psi^5},\ -\frac{49\,\pi^4\,\alpha2}{24\,t^4\,\psi^{10}}-\frac{7\,\pi^2\,\alpha1}{8\,t^4\,\psi^8}-\frac{7}{8\,t^4\,\psi^6}\right\}\right\}$$

Compute the number and energy densities:

```
ρEDER = Expand[D[pEDER, ψ] / t]
```

$$\left\{-\frac{1}{6}+\frac{1}{3}\pi^2\,t^2\,\alpha1+\frac{t\,\psi}{2}+\frac{1}{3}\pi^2\,t^3\,\alpha1\,\psi+t^2\,\psi^2+\frac{t^3\,\psi^3}{3},\right.$$
$$\left.\frac{7\,\pi^4\,\alpha2}{120\,t\,\psi^5}+\frac{\pi^2\,\alpha1}{24\,t\,\psi^3}+\frac{1}{8\,t\,\psi},\ -\frac{7\,\pi^4\,\alpha2}{24\,t\,\psi^6}-\frac{\pi^2\,\alpha1}{8\,t\,\psi^4}-\frac{1}{8\,t\,\psi^2},\ \frac{49\,\pi^4\,\alpha2}{48\,t\,\psi^7}+\frac{7\,\pi^2\,\alpha1}{24\,t\,\psi^5}+\frac{7}{48\,t\,\psi^3}\right\}$$

```
coefz2 = t^3 ψ^3 / 3
```

$$\frac{t^3\,\psi^3}{3}$$

```
ρEDERsimp = {coefz2, Expand[ρEDER / coefz2]}
```

$$\left\{\frac{t^3\,\psi^3}{3},\ \left\{1-\frac{1}{2\,t^3\,\psi^3}+\frac{\pi^2\,\alpha1}{t\,\psi^3}+\frac{3}{2\,t^2\,\psi^2}+\frac{\pi^2\,\alpha1}{\psi^2}+\frac{3}{t\,\psi},\right.\right.$$
$$\left.\left.\frac{7\,\pi^4\,\alpha2}{40\,t^4\,\psi^8}+\frac{\pi^2\,\alpha1}{8\,t^4\,\psi^6}+\frac{3}{8\,t^4\,\psi^4},\ -\frac{7\,\pi^4\,\alpha2}{8\,t^4\,\psi^9}-\frac{3\,\pi^2\,\alpha1}{8\,t^4\,\psi^7}-\frac{3}{8\,t^4\,\psi^5},\ \frac{49\,\pi^4\,\alpha2}{16\,t^4\,\psi^{10}}+\frac{7\,\pi^2\,\alpha1}{8\,t^4\,\psi^8}+\frac{7}{16\,t^4\,\psi^6}\right\}\right\}$$

```
εEDER = Expand[Simplify[t D[pEDER, t] - pEDER + ρEDER]]
```

$$\left\{-\frac{1}{6}+\frac{5}{12}\pi^2\,t^2\,\alpha1+\frac{7}{60}\pi^4\,t^4\,\alpha2+\frac{t\,\psi}{2}+\pi^2\,t^3\,\alpha1\,\psi+\frac{5\,t^2\,\psi^2}{4}+\frac{1}{2}\pi^2\,t^4\,\alpha1\,\psi^2+t^3\,\psi^3+\frac{t^4\,\psi^4}{4},\right.$$
$$\frac{19}{96}+\frac{7\,\pi^4\,\alpha2}{120\,t\,\psi^5}+\frac{7\,\pi^4\,\alpha2}{480\,\psi^4}+\frac{\pi^2\,\alpha1}{24\,t\,\psi^3}+\frac{\pi^2\,\alpha1}{48\,\psi^2}+\frac{1}{8\,t\,\psi}+\frac{1}{8}\mathrm{Log}\left[\frac{1}{t}\right]-\frac{1}{8}\mathrm{Log}[2\,\psi],$$
$$\left.-\frac{7\,\pi^4\,\alpha2}{24\,t\,\psi^6}-\frac{7\,\pi^4\,\alpha2}{120\,\psi^5}-\frac{\pi^2\,\alpha1}{8\,t\,\psi^4}-\frac{\pi^2\,\alpha1}{24\,\psi^3}-\frac{1}{8\,t\,\psi^2}-\frac{1}{8\,\psi},\ \frac{49\,\pi^4\,\alpha2}{48\,t\,\psi^7}+\frac{49\,\pi^4\,\alpha2}{288\,\psi^6}+\frac{7\,\pi^2\,\alpha1}{24\,t\,\psi^5}+\frac{7\,\pi^2\,\alpha1}{96\,\psi^4}+\frac{7}{48\,t\,\psi^3}+\frac{7}{96\,\psi^2}\right\}$$

```
coefz3 = ψ^4 t^4 / 4
```

$$\frac{t^4\,\psi^4}{4}$$

```
{coefz3, Expand[εEDER / coefz3]}
```

$$\left\{\frac{t^4\psi^4}{4},\left\{1-\frac{2}{3t^4\psi^4}+\frac{5\pi^2\alpha 1}{3t^2\psi^4}+\frac{7\pi^4\alpha 2}{15\psi^4}+\frac{2}{t^3\psi^3}+\frac{4\pi^2\alpha 1}{t\psi^3}+\frac{5}{t^2\psi^2}+\frac{2\pi^2\alpha 1}{\psi^2}+\frac{4}{t\psi},\right.\right.$$
$$\frac{7\pi^4\alpha 2}{30t^5\psi^9}+\frac{7\pi^4\alpha 2}{120t^4\psi^8}+\frac{\pi^2\alpha 1}{6t^5\psi^7}+\frac{\pi^2\alpha 1}{12t^4\psi^6}+\frac{1}{2t^5\psi^5}+\frac{19}{24t^4\psi^4}+\frac{\mathrm{Log}\left[\frac{1}{t}\right]}{2t^4\psi^4}-\frac{\mathrm{Log}[2\psi]}{2t^4\psi^4},$$
$$-\frac{7\pi^4\alpha 2}{6t^5\psi^{10}}-\frac{7\pi^4\alpha 2}{30t^4\psi^9}-\frac{\pi^2\alpha 1}{2t^5\psi^8}-\frac{\pi^2\alpha 1}{6t^4\psi^7}-\frac{1}{2t^5\psi^6}-\frac{1}{2t^4\psi^5},$$
$$\left.\left.\frac{49\pi^4\alpha 2}{12t^5\psi^{11}}+\frac{49\pi^4\alpha 2}{72t^4\psi^{10}}+\frac{7\pi^2\alpha 1}{6t^5\psi^9}+\frac{7\pi^2\alpha 1}{24t^4\psi^8}+\frac{7}{12t^5\psi^7}+\frac{7}{24t^4\psi^6}\right\}\right\}$$

Massless pair formulas

- **The expression for the chemical potential 'nu' in terms of n, g, T:**

```
sqt = Sqrt[729 n^2 + 3 g^2 π^2 T^6]
```

$$\sqrt{729 n^2+3 g^2\pi^2 T^6}$$

```
cbt = (-27 n + sqt)^(1/3)
```

$$\left(-27 n+\sqrt{729 n^2+3 g^2\pi^2 T^6}\right)^{1/3}$$

```
nu = (g π^4 / 3)^(1/3) T^2 / cbt - (π^2 / 9 / g)^(1/3) cbt
```

$$\frac{g^{1/3}\pi^{4/3}T^2}{3^{1/3}\left(-27 n+\sqrt{729 n^2+3 g^2\pi^2 T^6}\right)^{1/3}}-\left(\frac{1}{g}\right)^{1/3}\left(\frac{\pi}{3}\right)^{2/3}\left(-27 n+\sqrt{729 n^2+3 g^2\pi^2 T^6}\right)^{1/3}$$

- **Rephrase cbt in terms of a convenient variable α :**

```
cbt2 = 3 n^(1/3) (-1 + Sqrt[Simplify[(729 n^2 + 3 g^2 π^2 T^6) / 729 / n^2]])^(1/3)
```

$$3 n^{1/3}\left(-1+\sqrt{1+\frac{g^2\pi^2T^6}{243 n^2}}\right)^{1/3}$$

```
cbt3 = 3 n^(1/3) (-1 + √(1 + α))^(1/3)
```

$$3 n^{1/3}\left(-1+\sqrt{1+\alpha}\right)^{1/3}$$

```
α0 = g^2 π^2 T^6 / 243 / n^2
```

$$\frac{g^2\pi^2T^6}{243 n^2}$$

- **Rewrite 'nu' completely in terms of α**

```
cbt4 = (-1 + Sqrt[1 + α])^(1/3) / α^(1/6)
```

$$\frac{\left(-1+\sqrt{1+\alpha}\right)^{1/3}}{\alpha^{1/6}}$$

```
nu2 = π T / Sqrt[3] (1 / cbt4 - cbt4)
```

$$\frac{\pi\, T\left(\frac{\alpha^{1/6}}{\left(-1+\sqrt{1+\alpha}\right)^{1/3}}-\frac{\left(-1+\sqrt{1+\alpha}\right)^{1/3}}{\alpha^{1/6}}\right)}{\sqrt{3}}$$

Demonstrate that these are the same:

```
{N[nu /. g → 1 /. n → 2 /. T → 3], N[nu2 /. α → α0 /. g → 1 /. n → 2 /. T → 3]}
```

{1.30813, 1.30813}

```
{N[nu /. g → 3 /. n → 1 /. T → 2], N[nu2 /. α → α0 /. g → 3 /. n → 1 /. T → 2]}
```

{0.496892, 0.496892}

```
{N[nu /. g → 2 /. n → 3 /. T → 1], N[nu2 /. α → α0 /. g → 2 /. n → 3 /. T → 1]}
```

{3.73228, 3.73228}

```
{N[nu /. g → 1 /. n → 3 /. T → 2], N[nu2 /. α → α0 /. g → 1 /. n → 3 /. T → 2]}
```

{3.45517, 3.45517}

```
{N[nu /. g → 3 /. n → 2 /. T → 1], N[nu2 /. α → α0 /. g → 3 /. n → 2 /. T → 1]}
```

{2.47111, 2.47111}

```
{N[nu /. g → 2 /. n → 1 /. T → 3], N[nu2 /. α → α0 /. g → 2 /. n → 1 /. T → 3]}
```

{0.332918, 0.332918}

- **Some series expansions:**

```
Series[nu2 Sqrt[3] / π / T, {α, 0, 3}]
```

$$\frac{2^{1/3}}{\alpha^{1/6}}-\frac{\alpha^{1/6}}{2^{1/3}}+\frac{\alpha^{5/6}}{6\ 2^{2/3}}+\frac{\alpha^{7/6}}{12\ 2^{1/3}}-\frac{\alpha^{11/6}}{18\ 2^{2/3}}-\frac{5\,\alpha^{13/6}}{144\ 2^{1/3}}+\frac{77\,\alpha^{17/6}}{2592\ 2^{2/3}}+O[\alpha]^{19/6}$$

```
Series[nu2 Sqrt[3] / π / T, {α, ∞, 3}]
```

$$\frac{2\sqrt{\frac{1}{\alpha}}}{3}-\frac{8}{81}\left(\frac{1}{\alpha}\right)^{3/2}+\frac{32}{729}\left(\frac{1}{\alpha}\right)^{5/2}+O\left[\frac{1}{\alpha}\right]^{7/2}$$