Degenerate and non-degenerate expansions for the number density, energy density, and entropy of a non-relativistic fermi gas. The substitions u=p^2/2/m^{*}/T and y=-mu/T have been used to simplify the notation. The degenerate expansions are Sommerfeld expansions which results in an asymptotic (not convergent) series.

```
> restart;
```

Number density

```
> n:=Int(u^(1/2)/(1+exp(u+y)),u=0..infinity);
```

$$n := \int_0^\infty \frac{\sqrt{u}}{1+\mathbf{e}^{(u+y)}}\,du$$

Degenerate expansion (y->-\infty):

```
> f:=sqrt(u);
```

$$f := \sqrt{u}$$

```
> b:=int(f,u=0..-y)+Pi^2/6*subs(u=-y,diff(f,u));
```

$$b := \frac{2}{3}(-y)^{(3/2)} + \frac{\frac{1}{12}\pi^2}{\sqrt{-y}}$$

```
> c:=int(f,u=0..-y)+Pi^2/6*subs(u=-y,diff(f,u))+7*Pi^4/360*subs(u=-y
  ,diff(f,u$3));
```

$$c := \frac{2}{3}(-y)^{(3/2)} + \frac{\frac{1}{12}\pi^2}{\sqrt{-y}} + \frac{\frac{7}{960}\pi^4}{(-y)^{(5/2)}}$$

```
> evalf(subs(y=-30,[log10(abs(n-b)),log10(abs(n-c))]));
```

$$[-3.836985105, -5.834327409]$$

Non-degenerate expansion (y->\infty):

```
> ff:=convert(subs(zz=exp(y),series(1/(a+zz),zz=infinity)),polynom);
```

$$ff := \frac{1}{\mathbf{e}^y} - \frac{a}{(\mathbf{e}^y)^2} + \frac{a^2}{(\mathbf{e}^y)^3} - \frac{a^3}{(\mathbf{e}^y)^4} + \frac{a^4}{(\mathbf{e}^y)^5}$$

```
> d:=expand(int(sqrt(u)*exp(-u)*subs(a=exp(-u),ff),u=0..infinity));
```

$$d := \frac{1}{50}\frac{\sqrt{\pi}\sqrt{5}}{(\mathbf{e}^y)^5} + \frac{\frac{1}{2}\sqrt{\pi}}{\mathbf{e}^y} - \frac{1}{16}\frac{\sqrt{\pi}}{(\mathbf{e}^y)^4} + \frac{\frac{1}{18}\sqrt{\pi}\sqrt{3}}{(\mathbf{e}^y)^3} - \frac{1}{8}\frac{\sqrt{\pi}\sqrt{2}}{(\mathbf{e}^y)^2}$$

```
> evalf(subs(y=4,log10(abs(n-d))));
```

$$-11.55284197$$

Energy density

```
> epsilon:=Int(u^(3/2)/(1+exp(u+y)),u=0..infinity);
```

$$\varepsilon := \int_0^\infty \frac{u^{(3/2)}}{1+\mathbf{e}^{(u+y)}}\,du$$

Degenerate:

```
> f:=u^(3/2);
```

$$f := u^{(3/2)}$$

```
> b:=int(f,u=0..-y)+Pi^2/6*subs(u=-y,diff(f,u));
```

$$b := \frac{2}{5}(-y)^{(5/2)} + \frac{1}{4}\pi^2\sqrt{-y}$$

```
> c:=int(f,u=0..-y)+Pi^2/6*subs(u=-y,diff(f,u))+7*Pi^4/360*subs(u=-y
  ,diff(f,u$3));
```

$$c := \frac{2}{5}(-y)^{(5/2)} + \frac{1}{4}\pi^2\sqrt{-y} - \frac{7}{960}\frac{\pi^4}{(-y)^{(3/2)}}$$

```
> evalf(subs(y=-40,[log10(abs(epsilon-b)),log10(abs(epsilon-c))]));
```

[-2.550558629, -5.145509591]

Non-degenerate:

```
> d:=expand(int(u^(3/2)*exp(-u)*subs(a=exp(-u),ff),u=0..infinity));
```

$$d := -\frac{3}{128}\frac{\sqrt{\pi}}{(\mathbf{e}^y)^4} + \frac{\frac{3}{4}\sqrt{\pi}}{\mathbf{e}^y} + \frac{\frac{3}{500}\sqrt{\pi}\sqrt{5}}{(\mathbf{e}^y)^5} - \frac{3}{32}\frac{\sqrt{\pi}\sqrt{2}}{(\mathbf{e}^y)^2} + \frac{\frac{1}{36}\sqrt{\pi}\sqrt{3}}{(\mathbf{e}^y)^3}$$

```
> evalf(subs(y=4,log10(abs(epsilon-d))));
```

-10.98827312

Entropy

```
> s:=Int(sqrt(u)*(ln(1+exp(u+y))/(1+exp(u+y))+
  ln(1+exp(-u-y))/(1+exp(-u-y))),u=0..infinity);
```

$$s := \int_0^\infty \sqrt{u}\left(\frac{\ln(1+\mathbf{e}^{(u+y)})}{1+\mathbf{e}^{(u+y)}} + \frac{\ln(1+\mathbf{e}^{(-u-y)})}{1+\mathbf{e}^{(-u-y)}}\right)du$$

An alternate expression for the entropy is:

```
> salt:=Int(sqrt(u)*(log(1+exp(-y-u))+(u+y)/(1+exp(u+y))),u=0..infin
  ity);
```

$$salt := \int_0^\infty \sqrt{u}\left(\ln(1+\mathbf{e}^{(-u-y)}) + \frac{u+y}{1+\mathbf{e}^{(u+y)}}\right)du$$

Degenerate:

```
> f1:=sqrt(u);
```

$$f1 := \sqrt{u}$$

```
> c:=int(f1,u=0..-y)+Pi^2/6*subs(u=-y,diff(f1,u))+7*Pi^4/360*subs(u=
  -y,diff(f1,u$3));
```

$$c := \frac{2}{3}(-y)^{(3/2)} + \frac{\frac{1}{12}\pi^2}{\sqrt{-y}} + \frac{\frac{7}{960}\pi^4}{(-y)^{(5/2)}}$$

```
> s1:=-int(c,y);
```

$$s1 := \frac{4}{15}(-y)^{(5/2)} + \frac{1}{6}\pi^2\sqrt{-y} - \frac{7}{1440}\frac{\pi^4}{(-y)^{(3/2)}}$$

```
> f2:=sqrt(u)*(u+y);
```

$$f2 := \sqrt{u}\,(u+y)$$

```
> s2:=int(f2,u=0..-y)+Pi^2/6*subs(u=-y,diff(f2,u))+7*Pi^4/360*subs(u
  =-y,diff(f2,u$3));
```

$$s2 := -\frac{4}{15}\sqrt{-y}\,y^2 + \frac{1}{6}\pi^2\sqrt{-y} - \frac{7}{480}\frac{\pi^4}{(-y)^{(3/2)}}$$

```
> snew:=s1+s2;
```

$$snew := \frac{4}{15}(-y)^{(5/2)} + \frac{1}{3}\pi^2\sqrt{-y} - \frac{7}{360}\frac{\pi^4}{(-y)^{(3/2)}} - \frac{4}{15}\sqrt{-y}\,y^2$$

```
> snew2:=4/15*(-y)^(5/2)+1/3*Pi^2*sqrt(-y)-4/15*sqrt(-y)*y^2;
```

$$snew2 := \frac{4}{15}(-y)^{(5/2)} + \frac{1}{3}\pi^2\sqrt{-y} - \frac{4}{15}\sqrt{-y}\,y^2$$

```
> evalf(subs(y=-40,[log10(abs(s-snew)),log10(abs(s-snew2))]));
```

$$[-4.556610343, -2.124087775]$$

Non-degenerate:

```
> d:=expand(int(sqrt(u)*exp(-u)*(u+y)*subs(a=exp(-u),ff),u=0..infini
  ty));
```

$$d := \frac{3}{500}\frac{\sqrt{\pi}\sqrt{5}}{(\mathbf{e}^y)^5} - \frac{1}{8}\frac{\sqrt{\pi}\,y\sqrt{2}}{(\mathbf{e}^y)^2} - \frac{3}{32}\frac{\sqrt{\pi}\sqrt{2}}{(\mathbf{e}^y)^2} - \frac{1}{16}\frac{\sqrt{\pi}\,y}{(\mathbf{e}^y)^4} + \frac{\frac{1}{50}\sqrt{\pi}\sqrt{5}\,y}{(\mathbf{e}^y)^5} + \frac{\frac{1}{2}\sqrt{\pi}\,y}{\mathbf{e}^y} + \frac{\frac{3}{4}\sqrt{\pi}}{\mathbf{e}^y}$$
$$+ \frac{\frac{1}{36}\sqrt{\pi}\sqrt{3}}{(\mathbf{e}^y)^3} + \frac{\frac{1}{18}\sqrt{\pi}\,y\sqrt{3}}{(\mathbf{e}^y)^3} - \frac{3}{128}\frac{\sqrt{\pi}}{(\mathbf{e}^y)^4}$$

```
> d2:=subs(eps=exp(-y),convert(series(Int(sqrt(u)*log(1+eps*exp(-u))
  ,u=0..infinity),eps),polynom));
```

$$d2 := \frac{1}{2}\sqrt{\pi}\,\mathbf{e}^{(-y)} - \frac{1}{16}\sqrt{2}\sqrt{\pi}\,(\mathbf{e}^{(-y)})^2 + \frac{1}{54}\sqrt{3}\sqrt{\pi}\,(\mathbf{e}^{(-y)})^3 - \frac{1}{128}\sqrt{4}\sqrt{\pi}\,(\mathbf{e}^{(-y)})^4 + \frac{1}{250}\sqrt{5}\sqrt{\pi}\,(\mathbf{e}^{(-y)})^5$$

```
> dtot:=d+d2;
```

$$dtot := \frac{3}{500}\frac{\sqrt{\pi}\sqrt{5}}{(\mathbf{e}^{y})^{5}} - \frac{1}{8}\frac{\sqrt{\pi}\,y\sqrt{2}}{(\mathbf{e}^{y})^{2}} - \frac{3}{32}\frac{\sqrt{\pi}\sqrt{2}}{(\mathbf{e}^{y})^{2}} - \frac{1}{16}\frac{\sqrt{\pi}\,y}{(\mathbf{e}^{y})^{4}} + \frac{\frac{1}{50}\sqrt{\pi}\sqrt{5}\,y}{(\mathbf{e}^{y})^{5}} + \frac{\frac{1}{2}\sqrt{\pi}\,y}{\mathbf{e}^{y}} + \frac{\frac{3}{4}\sqrt{\pi}}{\mathbf{e}^{y}}$$
$$+ \frac{\frac{1}{36}\sqrt{\pi}\sqrt{3}}{(\mathbf{e}^{y})^{3}} + \frac{\frac{1}{18}\sqrt{\pi}\,y\sqrt{3}}{(\mathbf{e}^{y})^{3}} - \frac{3}{128}\frac{\sqrt{\pi}}{(\mathbf{e}^{y})^{4}} + \frac{1}{2}\sqrt{\pi}\,\mathbf{e}^{(-y)} - \frac{1}{16}\sqrt{2}\sqrt{\pi}\,(\mathbf{e}^{(-y)})^{2} + \frac{1}{54}\sqrt{3}\sqrt{\pi}\,(\mathbf{e}^{(-y)})^{3}$$
$$- \frac{1}{128}\sqrt{4}\sqrt{\pi}\,(\mathbf{e}^{(-y)})^{4} + \frac{1}{250}\sqrt{5}\sqrt{\pi}\,(\mathbf{e}^{(-y)})^{5}$$

```
> evalf(subs(y=4,log10(abs(s-dtot))));
```

-10.63637897

```
>
```